


# Lathe Turning Software Tool, version 2.5

By R. G. Sparber

Copyright protects this document.<sup>1</sup>



I've never been very good at concentrating on two things at the same time. Running my lathe is within my limited mental abilities as long as I don't have to do a lot of math and reasoning at the same time.

But part of accurate turning is the need to measure the current diameter, calculate how far to go in order to reach my "goal" diameter, and then feed in a reasonable amount for my next cut. Between cuts it is a good idea to monitor the relationship between dialed in cut and actual cut so corrections can be made.



The process repeats until I perfectly hit the goal (not very often) or I blow past the goal diameter.

---

<sup>1</sup> You are free to copy and distribute this document but not change it.



It is standard practice, if you own a digital caliper, to zero the display at the diameter you want to reach. Here I have zeroed the caliper with the jaws closed and then moved them apart until I read 1.100".



Then I again zero the display.



When I measure the current diameter, I read the current minus the final diameter. Here you see my part is 0.214" larger in diameter than my final.

Divide the reading by 2 to find the total tool in-feed for an outside diameter cut (or out-feed for boring ) to reach the final diameter. In this example, I get 0.107". Simple enough. Yet I seem to often get turned around and continue to make passes after the display shows a sign change. This sign change means I missed my target. I've long since given up on even dividing by 2 in my head while running my lathe since it is the reason I spoil parts all too often.

**If you have no problem with the this standard practice, there isn't much reason to read further.**

My solution to the distraction of doing this math while making chips is to write a program for my HP 35S programmable calculator. I was using this calculator to figure my in-feed anyway so it isn't much of a change in procedure. A video of the program in action can be seen at

<http://youtu.be/49vgfYtB3Gs>

I found that placing the calculator in a high quality "Hefty® OneZip®" bag kept oil and swarf out.

## **"Feature Creep"**

This program started out very small since it just told me the diameter to go before reaching my goal. But over time I saw new things for it to do and it finally got too big for my HP32S. After re-entering the program on my much newer HP35S, I was able to add even more features. The present program takes a lot of effort to type into the calculator. It even takes some time to answer all of the questions posed by the program. But once running, I hope you will find that it supplies useful and accurate data that will guide you to perfect IDs and ODs every time.


## Program Overview

The program is designed to assist the user by specifying the cross slide dial position given measured diameters and desired Depth Of Cut (DOC). It works for both OD turning and ID boring. For the case of an OD turning operation, here is the idea: The user specifies the goal diameter. Then they tell the program the current diameter.

The program tells the user how much in-feed is needed to reach the goal diameter. The rest of the user/program dialog involves feeding in the cutter and measuring the current diameter until the goal diameter is reached or passed.

## Background

I have tried to comment each line of the program in enough detail so the logic can



be implemented on other such calculators. It should also be clear to those using metric how to modify the program so all displays look right for them.

My lathe's cross slide dial is marked to show radius. If your lathe shows diameter, some adjustments to the program will be needed.

The program is set up so most displayed information stays on the screen until the user wants to move on by pressing Run/Stop (R/S)<sup>2</sup>.

---

<sup>2</sup> This feature can be disabled by adding a pause (PSE) after each prompt.

## Program "Dialog"

The conversation between user and program is the same for turning an ID or an OD. I will present the OD turning case.

Step	Prompt	User Action	example	comment
1		Start program by pressing XEQ A followed by Enter.		
2	L ASSIST RS	press R/S		Program is called "Lathe Assist". RS reminds user to press R/S
3	DIAL AT NRS	Prompt for present position of the dial. Then press R/S.	0 if set at touchdown.	The N in NRS reminds user to enter a Number. Top displayed number is thou per revolution of dial. Change by modifying program. Bottom number is value just entered.
4	GOAL DIA NRS	Prompt for desired final diameter. Press R/S.	1	Top number is present dial setting. Bottom number is value just entered.
5	DIA NOW RS	Prompt for present measured diameter. Press R/S.	1.1	Top number is goal diameter. Bottom number is value just entered.
6	R TO GO RS	Program is about to tell you how much radius is left to cut. Press R/S.		
7	X= xx.x	Program displays total radius left to cut in tenths. Press R/S.	50.0	Always positive until goal missed
8	NEXT DOC NRS	Prompt for next depth of cut. Press R/S.	25	Top number is radius to go. Bottom number is value just entered.

9	SET DIAL	Program is about to tell you where to set your cross feed dial. Press R/S.	25.	Only displays for a moment.
10	DIA NOW NRS	Prompt for measured diameter after cut. Press R/S.	1.048	Top number is goal diameter. Bottom number is value just entered.
11	DOC CORRECTION	Program is about to tell you the depth of cut correction factor.		Only displays for a moment.
12	X= x.xxx	Program displays depth of cut correction factor. Press R/S	0.962	Ideal case of actual DOC equals dialed in DOC shows 1.000.
	Return to step 6	Preparing for next cut.		

When R TO GO is zero or negative, the goal diameter has been reached or has been missed. In either case, the program prompts for a new GOAL DIA.

Entering 0 at any prompt except dial position sends you to the start of the program.

## Using the Correction Factor

The correction factor is intended to cancel fixed error when a series of almost equal depth cuts are made. I use it by first dialing in a DOC and taking a pass. Then I dial in the same DOC and take a second pass. Only then do I measure the diameter. This is followed by a third cut at the same DOC. I calculate the change in radius between these last two cuts. In this way I attempt to put the same stress on the cutter during each pass which should reduce error due to varying pressure.

Say I am cutting an OD and feed in .005" twice. I measure the actual reduction in radius between the last two cuts and get .005". In this case, my correction factor will be 1. But what if I get .004"? If this difference is due to fixed errors, then every time I dial in .005" I can expect to see a change in radius of .004". My correction factor would be

$$\text{correction factor} = \frac{\text{dialed in depth of cut}}{\text{actual change in radius}}$$

$$\text{correction factor} = \frac{5 \text{ thou}}{4 \text{ thou}} = 1.25$$

Maximum accuracy of this correction factor is when dialing in a .005" value. But it would still be useful near this amount.

Say I wanted to reduce the radius by .005". I can estimate the correct change in the dial by multiplying my desired change in radius by my correction factor.

$$\text{estimated dial change} = \text{desired depth of cut} \times \text{correction factor}$$

$$\text{estimated dial change} = 5 \text{ thou} \times 1.25$$

$$\text{estimated dial change} = 6 \text{ thou}$$

So if I dial in 6 thou, I will get an actual reduction in the radius of 5 thou *if* all error is constant. In the real world we do have both fixed and random error. It is impossible to compensate for the random error. That is simply the limit of your machine.

## Turning Program

<b>LBL A</b>	start of program
L ASSIST RS	display program name and wait for user to press R/S
SF 10	input as SF.0 - this lets me display text prompts
SF 0	flag 0 set to true which means first current diameter not input yet
FIX 0	
100	DEFAULT DIAL SIZE
STO A	
DIAL AT NRS	dial position before first cut ;Y=DIAL SIZE, X is user input from DIAL AT; NRS means input number and then R/S
STO B	
FIX 3	set display to the nearest thou
GOAL DIA NRS	prompt for goal diameter
STO G	
X=0?	
GTO A001	
<b>LBL B</b>	top of cutting iterations logic
RCL M	recall current diameter (is left over from previous run on first pass)
STO N	save current diameter in N as previous current diameter to be used for DOC calculation
FIX 3	set display to the nearest thou
RCL G	put GOAL DIA in stack so when DIA NOW displayed, Y will be the goal diameter
DIA NOW NRS	prompt for current diameter
STO M	
X=0?	
GTO A001	
FS?0	if this is the first pass, skip DOC correction calculation
GTO F001	
RCL N	calculate actual DOC: (previous diameter - current



	diameter)/2 with result in decimal inches
RCL M	
-	
2	
/	
RCL S	sign correction so RADIUS TO GO is always positive unless goal missed
*	
1000	convert actual DOC to thou
*	
X=0?	Divide by 0 check
GTO A001	if actual DOC zero, warn and restart program
RCL D	recall previous dialed DOC since this is not the first pass; is in thou
Swap X and Y	
/	now have dialed in DOC/ actual DOC
ABS	force correction factor to be positive for ID and OD
FIX 3	set display to 3 places past decimal
DOC CORRECTION	displayed for 1 second
PSE	
STO X	temporary storage in X so I can use VIEW
VIEW X	display DOC correction factor
<b>LBL F</b>	remaining calculation of RADIUS TO GO
RCL G	bring back goal diameter
RCL M	bring back current diameter
X=0?	If current diameter is 0, it means user wants to set a new goal diameter
GTO A001	so restart the program; otherwise
-	subtract current from goal diameters
FS? 0	if this is the first pass with this goal diameter, figure out if this is an ID or OD and set variable S; otherwise calculate RADIUS TO GO.
<b>XEQ D001</b>	
-2	otherwise, convert to in or out feed; neg sign inverts sign of S for correct sign of RADIUS TO GO
/	
RCL S	recall sign correction variation and apply
*	

1000	prepare to display in thou
*	
R TO GO RS	tell user the in or out feed value is about to be displayed
X<0?	If value is negative, user missed the goal so display message and then value
<b>XEQ Y001</b>	
FIX 1	to display tenths
STO X	temporary storage in X
VIEW X	display radius left to go (negative means you missed it)
X<0?	If goal missed, prompt for new goal
GTO A001	restart program
X=0?	If goal exactly reached, prompt for new goal
GTO A001	restart program
FIX 0	to display thou
NEXT DOC NRS	otherwise, prompt for next dialed in depth of cut
STO D	
X=0?	
GTO A001	
	<b>calculate new dial position before cut made</b>
RCL B	recall current dial value
RCL D	recall NEXT DOC
RCL S	recall feed sign correction
*	change sign of NEXT DOC depending on OD or ID cut
+	add from past dial position to get next dial position
	If OD, dial numbers increase; if ID, they decrease
STO C	store next dial setting but may need adjustment first
<b>LBL H</b>	
RCL C	recall next dial setting
RCL A	test if next dial setting is greater than max dial number
X<Y?	is next dial setting greater than max dial number?
GTO C001	if it is, subtract max dial number and retest
RCL C	recall next dial position again
X<0?	Is next dial position negative?
GTO E001	if it is, add max dial number and retest
FIX 0	
SET DIAL	display new dial position for 1 second
PSE	
STO X	save to X so I can use the VIEW function
VIEW X	wait for user to read new dial position and hit R/S
STO B	next dial setting is now current dial setting

GTO B001	prepare for next current diameter
<b>LBL C</b>	next dial position is larger than max dial value so Reduce it by the max dial number and retest
RCL C	recall next dial setting which we know is too big
RCL A	recall max dial value
-	
STO C	next dial setting is now reduced by one dial revolution
GTO H001	retest next dial setting to see if it is a value on the dial
<b>LBL E</b>	next dial position is negative so add max dial number and retest
RCL C	recall next dial setting which is negative
RCL A	recall max dial value
+	
STO C	next dial setting is now increased by one dial revolution
GTO H001	retest next dial setting to see if it is a value on the dial
<b>LBL D</b>	just before first cut, determine if this is an ID or OD turning operation; input is goal minus first cut diameter
X<0?	If negative, this is an OD, go to subroutine E to set sign correction value to +1 and clear flag that says this is the first pass
GTO G001	
-1	otherwise, Goal minus first cut it is positive so this is an ID; change sign correction value to -1; value in X moved to Y
<b>LBL Z</b>	
STO S	S will control the sign of the RADIUS TO GO value so negative will mean over
CF 0	we are no longer at first iteration so do not set sign correction value again
Roll down	move value that was in X at start of subroutine back to X
RTN	
<b>LBL G</b>	<b>only used by subroutine D</b>
1	value in X moved to Y
GTO Z	return to subroutine D

**LBL Y**

MISSED BY  
RTN

## Checksums

These values will let you identify which program segments were input correctly. If the checksums match, all is well. Otherwise, look for an command input wrong.

Segment	Checksum
A	2F22
B	17F5
F	2E87
H	524B
C	68B4
E	B26F
D	CD50
Z	86A4
G	4BE9
Y	EC19

## Test Cases

Each test case exercises a different logical path in the program.

1. Input DIAL AT 0, goal diameter of 0 should restart the program.
2. Input DIAL AT 0, goal diameter of 1, DIA NOW set to 0 should restart the program.
3. Input DIAL AT 0, GOAL DIA of 1, DIA NOW set to 1 should show a R TO GO of 0 and then the program restarts.
4. Input DIAL AT 0, GOAL DIA of 1, DIAMETER NOW set to 2 should cause RADIUS TO GO to be 500.0 (this is an OD being turned). Set NEXT DOC to 50 and user is told to SET DIAL to 50. Input DIA NOW of 1.9. DOC correction should be 1.000. R TO GO should be 450.0. Set NEXT DOC to 0 and see restart of program.
5. Input DIAL AT 0, GOAL DIA of 2, DIA NOW set to 1 should cause RADIUS TO GO to be 500.0 (this is an ID being turned). Set NEXT DOC to 50 and user is told to SET DIAL to 50. Input DIA NOW of 1.1. DOC correction should be 1.000. R TO GO should be 450.0. Set NEXT DOC to 0 and see restart of program.
6. Input DIAL AT 0, GOAL DIA of 1, DIA NOW set to 1.1. R TO GO is 50. Set NEXT DOC to 50. Program should instruct user to SET DIAL to 50. Input DIA NOW as .9. DOC Correction is 0.500. R TO GO is followed by MISSED BY and then -50.0. Program then restarts.

## Variable Definitions

Flag 0 - set to true before first cut and set false after first cut

Flag 10 - enables the display of text

A - maximum number on dial

B - current dial value

C - next dial setting

D - most recent dialed in depth of cut in thousandths of an inch

G - goal diameter

M - current diameter in decimal inches

N - previous current diameter in decimal inches

S - equals -1 if ID being cut or +1 if OD being cut; used to keep feed value positive until cut is past the goal

X - temporary storage used to display using the VIEW function

## Acknowledgements

Thanks to Gordon Haag for pointing out that some lathes have dials marked for diameter and not radius. Thanks to Nelson Collar, "CT2", David, and L.H. Garlinghouse for alerting me to typos.

I welcome your comments and questions.

Rick Sparber

[Rgsparber@aol.com](mailto:Rgsparber@aol.com)

[Rick.Sparber.org](http://Rick.Sparber.org)

